
ASP Tutorial

In our ASP tutorial you will learn about ASP, and how to execute scripts on your server. You will also see that ASP is the most powerful tool for making dynamic and interactive Web pages. Start Learning ASP!
ASP References

At W3Schools you will find complete ASP references about built-in objects and components, and their properties and methods. ASP References.

ASP Quick Reference

ASP Quick Reference from W3Schools. Print it, and fold it in your pocket. ASP Quick Reference
ASP Examples

Learn by 100 examples! Because ASP scripts are executed on the server, you can not view ASP code in a browser, you will only see the output from ASP which is plain HTML. At W3Schools every example displays the hidden ASP code. This will make it easier for you to understand how it works. ASP Examples!
ASP Quiz Test

Test your ASP skills at W3Schools! Start ASP Quiz! 

Introduction to ASP



An ASP file can contain text, HTML tags and scripts. Scripts in an ASP file are executed on the server



What you should already know

Before you continue you should have some basic understanding of the following:

· WWW, HTML and the basics of building Web pages 

· A scripting language like JavaScript or VBScript 

If you want to study these subjects, go to our Home Page


What is ASP?

· ASP stands for Active Server Pages 

· ASP is a program that runs inside IIS 

· IIS stands for Internet Information Services 

· IIS comes as a free component with Windows 2000 

· IIS is also a part of the Windows NT 4.0 Option Pack 

· The Option Pack can be downloaded from Microsoft 

· PWS is a smaller - but fully functional - version of IIS 

· PWS can be found on your Windows 95/98 CD 

ASP Compatibility

· ASP is a Microsoft Technology 

· To run IIS you must have Windows NT 4.0 or later 

· To run PWS you must have Windows 95 or later 

· ChiliASP is a technology that runs ASP without Windows OS 

· InstantASP is another technology that runs ASP without Windows 

What is an ASP File?

· An ASP file is just the same as an HTML file 

· An ASP file can contain text, HTML, XML, and scripts 

· Scripts in an ASP file are executed on the server 

· An ASP file has the file extension ".asp" 

How Does ASP Differ from HTML?

· When a browser requests an HTML file, the server returns the file 

· When a browser requests an ASP file, IIS passes the request to the ASP engine. The ASP engine reads the ASP file, line by line, and executes the scripts in the file. Finally, the ASP file is returned to the browser as plain HTML 



What can ASP do for you?

· Dynamically edit, change or add any content of a Web page 

· Respond to user queries or data submitted from HTML forms 

· Access any data or databases and return the results to a browser 

· Customize a Web page to make it more useful for individual users 

· The advantages of using ASP instead of CGI and Perl, are those of simplicity and speed 

· Provides security since your ASP code can not be viewed from the browser 

· Since ASP files are returned as plain HTML, they can be viewed in any browser 

· Clever ASP programming can minimize the network traffic 



Run ASP on Your PC



You can run ASP on your own PC without an external server. To do that, you must install Microsoft's Personal Web Server (PWS) or Internet Information Server (IIS) on your PC.



How to Run ASP on your own PC

You can run ASP on your own PC without an external server. To do that, you must install Microsoft's Personal Web Server (PWS) or Internet Information Server (IIS) on your PC.

If you are serious about using ASP, you should have at least Windows 98, Second Edition. 
If you are really serious about using ASP, you should go for Windows 2000.


How to install PWS and run ASP on Windows 95

Personal Web Server (PWS) is not shipped with Windows 95 !!

To run ASP on Windows 95, you will have to download "Windows NT 4.0 Option Pack" from Microsoft.

Download the "Windows NT 4.0 Option Pack"


How to install PWS and run ASP on Windows NT

Personal Web Server (PWS) is not shipped with Windows NT !!

To run ASP on Windows NT, you will have to download "Windows NT 4.0 Option Pack" from Microsoft.

Download the "Windows NT 4.0 Option Pack"


How to install PWS and run ASP on Windows 98

1. Open the Add-ons folder on your Windows98 CD, find the PWS folder and run the setup.exe file. 

2. An Inetpub folder will be created on your harddrive. Open it and find the wwwroot folder. 

3. Create a new folder, like "MyWeb", under wwwroot. 

4. Use a text editor to write some ASP code, save the file as "test1.asp" in the "MyWeb" folder. 

5. Make sure your Web server is running. 

6. Open your browser and type in "http://localhost/MyWeb/test1.asp", to view your first ASP page. 



How to install PWS and run ASP on Windows ME

Personal Web Server (PWS) is not included with Windows Me !!



How to install IIS and run ASP on Windows 2000

1. From your Start Button, go to Settings, and Control Panel 

2. In the Control Panel window select Add/Remove Programs 

3. In the Add/Remove window select Add/Remove Windows Components 

4. In the Wizard window check Internet Information Services, click OK 

5. An Inetpub folder will be created on your harddrive 

6. Open the Inetpub folder, and find a folder named wwwroot 

7. Create a new folder, like "MyWeb", under wwwroot. 

8. Use a text editor to write some ASP code, save the file as "test1.asp" in the "MyWeb" folder 

9. Make sure your Web server is running 

10. Open your browser and type in "http://localhost/MyWeb/test1.asp", to view your first ASP page 



How to install IIS and run ASP on Windows XP Professional

Note: You cannot run ASP on Windows XP Home Edition.

1. Insert the Windows XP Professional CD-Rom into your CD-Rom Drive 

2. From your Start Button, go to Settings, and Control Panel 

3. In the Control Panel window select Add/Remove Programs 

4. In the Add/Remove window select Add/Remove Windows Components 

5. In the Wizard window check Internet Information Services, click OK 

6. An Inetpub folder will be created on your harddrive 

7. Open the Inetpub folder, and find a folder named wwwroot 

8. Create a new folder, like "MyWeb", under wwwroot. 

9. Use a text editor to write some ASP code, save the file as "test1.asp" in the "MyWeb" folder 

10. Make sure your Web server is running 

11. Open your browser and type in "http://localhost/MyWeb/test1.asp", to view your first ASP page 



ASP Syntax



You cannot view ASP code in a browser, you will only see the output from ASP which is plain HTML. This is because the scripts are executed on the server before the result is sent to the browser.

In our school, every example displays the hidden ASP code. This will make it easier for you to understand how it works.



Examples 

Write text
How to write some text into the body of the HTML document with ASP.

<html>
<body>

<%
response.write("Hello World!")
%>

</body>
</html>

Text and HTML tags
How to format the text with HTML tags.

<html>
<body>

<%
response.write("<h2>Hello World!<br>This sentence uses HTML tags to format the text!</h2>")
%>

</body>
</html>



The Basic Syntax Rule 

An ASP file normally contains HTML tags, just as a standard HTML file. In addition, an ASP file can contain server scripts, surrounded by the delimiters <% and %>. Server scripts are executed on the server, and can contain any expressions, statements, procedures, or operators that are valid for the scripting language you use. 



The Response Object 

The Write method of the ASP Response Object is used to send content to the browser. For example, the following statement sends the text "Hello World" to the browser: Response.Write("Hello World"). 



VBScript

In ASP it is possible to use different scripting languages. The default scripting language in ASP is VBScript, as in this example:

	<html>

<body>

<%

response.write("Hello World!")

%>
</body>

</html>


The example above uses the response.write function to write Hello World! into the body of the HTML document. 



JavaScript 

To make JavaScript the default scripting language within a particular page, insert a language specification at the top of that page.

	<%@ language="javascript" %>

<html>

<body>

<%

Response.Write("Hello World!")

%>
</body>

</html>


Note that - unlike VBScript - JavaScript is case sensitive. You will have to write your ASP code with uppercase letters and lowercase letters when the language requires it. 



Other Scripting Languages 

ASP comes with VBScript and JScript, Microsoft's implementation of Javascript. If you want to script in another language, like PERL, REXX, or Python, you have to install scripting engines for them.  

Because the scripts are executed on the server, the browser that requests the ASP file does not need to support scripting. 



ASP Variables



A variable declared outside a procedure can be changed by any script in the ASP file. A variable declared inside a procedure, is created and destroyed every time the procedure is executed. 



Examples

Create a variable
Variables are used to store information. This example demonstrates how to create a variable, assign a value to it, and insert the variable value into a text.

<html>
<body>

<%
Dim name
name="Jan Egil"

response.write("My name is: " & name)
%>

</body>
</html>
Create an array
Arrays are used to store a series of related data items. This example demonstrates how you can make an array that stores names.

<html>
<body>
<%
Dim famname(5)
famname(0) = "Jan Egil"
famname(1) = "Tove"
famname(2) = "Hege"
famname(3) = "Stale"
famname(4) = "Kai Jim"
famname(5) = "Borge"
For i = 0 to 5
      response.write(famname(i) & "<br />")
Next
%>

</body>
</html>

Looping through headers
This example demonstrates how you can loop through the 6 headers in HTML.

<html>
<body>

<%
Dim i 

for i = 1 to 6
      response.write("<h" & i & ">This is header " & i & "</h" & i & ">")
next
%>

</body>
</html>

Time-based greeting using VBScript
How to write VBScript syntax in ASP. This example will display a different message to the user depending on the time on the server.

<html>

<body>
<%
Dim h
h = hour(now())

response.write("<p>" & now())
response.write(" (Norwegian Time) </p>")

If h <  12 then
      response.write("Good Morning!")
else
      response.write("Good day!")
end if
%>
</body>

</html>

Time-based greeting using JavaScript
How to write JavaScript syntax in ASP. This example is the same as the one above, but the syntax is different.

<%@ language="javascript" %>
<html>

<body>
<%
var d = new Date()
var h = d.getHours()

Response.Write("<p>")
Response.Write(d + " (Norwegian Time)")
Response.Write("</p>")

if (h <  12)
      {
      Response.Write("Good Morning!")
      }
else
      {
      Response.Write("Good day!")
      }
%>
</body>

</html>



Lifetime of Variables 

A variable declared outside a procedure can be accessed and changed by any script in the ASP page in which it is declared. 

A variable declared inside a procedure is created and destroyed every time the procedure is executed. No scripts outside that specific procedure can access or change that variable. 

To make a variable accessible to several ASP pages, declare it either as a session variable or as an application variable. 

Session Variables 

Session variables store information about one single user, and are available to all pages in one application. Common information stored in session variables are username and userid. To create a session variable, store it in a Session Object.

Application Variables 

Application variables are also available to all pages in one application. Application variables are used to hold information about all users in a specific application. To create an application variable, store it in an Application Object. 



ASP Procedures



In ASP you can call a JavaScript procedure from a VBScript and vice versa. 



Examples 

Call a procedure using VBScript
How to call a VBScript procedure from ASP.

<html>

<head>
<%
sub vbproc(num1,num2)
response.write(num1*num2)
end sub
%>
</head>

<body>
<p>
The result of the calculation is: <%call vbproc(3,4)%>
</p>
<p>
Or, you can call a procedure like this:
</p>
<p>
The result of the calculation is: <%vbproc 3,4%>
</p>
</body>

</html>

Call a procedure using JavaScript
How to call a JavaScript procedure from ASP.

<%@ language="javascript" %>
<html>

<head>
<%
function jsproc(num1,num2)
{
Response.Write(num1*num2)
}
%>
</head>

<body>
The result of the calculation is: <%jsproc(3,4)%>
</body>

</html>

Call procedures using VBScript
How to call a JavaScript procedure and a VBScript procedure from ASP.

<html>

<head>
<%
sub vbproc(num1,num2)
Response.Write(num1*num2)
end sub
%>

<script  language="javascript" runat="server">
function jsproc(num1,num2)
{
Response.Write(num1*num2)
}
</script>
</head>

<body>
The result of the calculation is: <%call vbproc(3,4)%><br><br>
The result of the calculation is: <%call jsproc(3,4)%>
</body>

</html>



Procedures 

ASP code can contain procedures and functions: 

	<html>

<head>

<%

sub vbproc(num1,num2)

response.write(num1*num2)

end sub

%>

</head>

<body>

The result of the calculation is: <%call vbproc(3,4)%>

</body>

</html>


Insert the <%@ language="language" %> line above the <html> tag if you want to write procedures or functions in a scripting language other than the default: 

	<%@ language="javascript" %>

<html>

<head>

<%

function jsproc(num1,num2)

{

Response.Write(num1*num2)

}

%>

</head>

<body>

The result of the calculation is: <%jsproc(3,4)%>

</body>

</html>




Calling a Procedure 

When calling a VBScript or a JavaScript procedure from an ASP page written in VBScript, you can use the "call" keyword followed by the procedure name. If a procedure requires parameters, the parameter list must be enclosed in parentheses when using the "call" keyword. If you omit the "call" keyword, the parameter list must not be enclosed in parentheses. If the procedure has no parameters, the parentheses are optional. 

When calling a JavaScript or a VBScript procedure from an ASP page written in JavaScript, always use parentheses after the procedure name.  



ASP Forms and User Input



You can get form information with the Request.QueryString command and the Request.Form command.



Examples

A form that uses the "get" method
This example demonstrates how to interact with the user, with the Request.QueryString command.

html>
<body>

<form action="demo_reqquery.asp" method="get">
Please type your first name:
<input type="text" name="fname"><br><br>
<input type="submit" value="Submit">
</form>

<%
If Request.QueryString("fname")<>"" Then
      Response.Write ("Hello " & Request.QueryString("fname") & "!")
      Response.Write ("<br>How are you today?")
End If
%>

</body>
</html>
A form that uses the "post" method
This example demonstrates how to interact with the user, with the Request.Form command.

<html>
<body>

<form action="demo_simpleform.asp" method="post">
Please type your first name:
<input type="text" name="fname">
<br>
<input type="submit" value="Submit">
</form>

<%
If Request.Form("fname")<>"" Then
      Response.Write ("Hello " & Request.Form("fname") & "!")
      Response.Write ("<br>How are you today?")
End If
%>

</body>
</html>

A form with radio buttons
This example demonstrates how to interact with the user, through radio buttons, with the Request.Form command.

<html>
<body>
<%
dim cars
cars=Request.Form("cars")
%>

<form action="demo_radiob.asp" method="post">
<p>Please select your favorite car:</p>
<input type="radio" name="cars"
<%if cars = "Volvo" then Response.Write("checked")%>
value="Volvo"> Volvo V70
<br>
<input type="radio" name="cars"
<%if cars = "Saab" then Response.Write("checked")%>
value="Saab"> Saab 95
<br>
<input type="submit" value="Submit">
</form>
<%
if cars<>"" then
   Response.Write("<p>Your favorite car is: " & cars & "</p>")
end if
%>

</body>
</html>


User Input 

To get information from forms, you can use the Request Object.

A simple form example:

	<form method="get" action="../pg.asp">

First Name:  <input type="text" name="fname"><br>

Last Name: <input type="text" name="lname"><br>

<input type="submit" value="Send">

</form>


There are two ways to get form information: The Request.QueryString command and the Request.Form command. 



Request.QueryString 

The Request.QueryString command collects the values in a form as text.  

Information sent from a form with the GET method is visible to everybody (in the address field). Remember that the GET method limits the amount of information to send. 

If a user typed "Bill" and "Gates" in the form example above, the url sent to the server would look like this:

	http://www.w3schools.com/pg.asp?fname=Bill&lname=Gates


The ASP file "pg.asp" contains the following script:

	<body>

Welcome

<%

response.write(request.querystring("fname"))

response.write("&nbsp;")

response.write(request.querystring("lname"))

%>

</body>


The example above writes this into the body of a document:

	Welcome Bill Gates




Request.Form 

To collect the values in a form with the POST method, use the Request.Form command. 

Information sent from a form with the POST method is invisible to others. The POST method has no limits, you can send a large amount of information.  

If a user typed "Bill" and "Gates" in the form example above, the url sent to the server would look like this: 

	http://www.w3schools.com/pg.asp


The ASP file "pg.asp" contains the following script:

	<body>

Welcome

<%

response.write(request.form("fname"))

response.write("&nbsp;")

response.write(request.form("lname"))

%>

</body>


The example above writes this into the body of a document:

	Welcome Bill Gates




Form Validation

The form input should be validated on the browser, by client side scripts. Browser validation has a faster response time, and reduces the load on the server.

You should consider using server validation if the input from a form is inserted into a database. A good way to validate the form on a server is to post the form into itself, instead of jumping to a different page. The user will then get the error messages on the same page as the form. This makes it easier to discover the error.



ASP Cookies



A cookie can be used to identify a user.



Examples 

Welcome cookie
How to create a Welcome cookie.

<%
response.cookies("NumVisits").Expires = date + 365 
num=request.cookies("NumVisits")

If num = "" Then
      response.cookies("NumVisits") = 1
Else
      response.cookies("NumVisits") = num + 1
End If
%>

<html>
<body>
<%
if num="" then
%>
Welcome! This is the first time you are visiting this Web page.
<%
else
%>
You have visited this Web page <%response.write(num) 
if num=1 then
      response.write " time before!"
else
      response.write " times before!"
end if
end if
%>

</body>
</html>



What is a Cookie? 

A cookie can be used to identify a user. It is a small file that the server embeds on the user's computer. Each time the same computer asks for a page through a browser, it will send the cookie too.

With ASP, you can both create cookies and retrieve the value of cookies.



How to Create a Cookie

The "Response.Cookies" command is used to create a cookie:

	<%

Response.Cookies("firstname")="Alex"

%>


In the code above, we have created a cookie named "firstname" and assigned the value "Alex" to it. 

Note: The Response.Cookies command must appear before the <html> tag. 

It is also possible to assign some properties to a cookie, like setting a date when a cookie should expire: 

	<%

Response.Cookies("firstname")="Alex" 

Response.Cookies("firstname").Expires=#May 10,2002#

%>


Now the cookie named "firstname" has the value of "Alex", and it will expire from the user's computer at May 10, 2002. 



How to Get a Cookie Value

The "Request.Cookies" command is used to get a cookie value. 

In the example below, we retrieve the value of the cookie "firstname" and display it on a page:

	<%

fname=Request.Cookies("firstname")

response.write("Firstname=" & fname)

%>



Output: 

Firstname=Alex



A Cookie with Keys 

A cookie can also contain a collection of multiple values. We say that the cookie has Keys. 

In the example below, we will create a cookie-collection named "user". The "user" cookie has Keys that contains information about a user:

	<%

Response.Cookies("user")("firstname")="John"

Response.Cookies("user")("lastname")="Smith"

Response.Cookies("user")("country")="Norway"

Response.Cookies("user")("age")="25"

%>




Read all Cookies

Look at the following code:

	<%

Response.Cookies("firstname")="Alex"

Response.Cookies("user")("firstname")="John"

Response.Cookies("user")("lastname")="Smith"

Response.Cookies("user")("country")="Norway"

Response.Cookies("user")("age")="25"

%>


Assume that your server has sent all the cookies above to a user.

Now we want to read all the cookies sent to a user. The example below shows how to do it (note that the code below checks if a cookie has Keys with the HasKeys property):

	<html>

<body>

<%

dim x,y

for each x in Request.Cookies

  response.write("<p>")

  if Request.Cookies(x).HasKeys then

    for each y in Request.Cookies(x)

      response.write(x & ":" & y & "=" & Request.Cookies(x)(y))

      response.write("<br />")

    next

  else

    Response.Write(x & "=" & Request.Cookies(x) & "<br />")

  end if

  response.write "</p>"

next

%>

</body>

</html>


Output:

firstname=Alex

user:firstname=John
user:lastname=Smith
user: country=Norway
user: age=25



What if a Browser Does NOT Support Cookies?

If your application deals with browsers that do not support cookies, you will have to use other methods to pass information from one page to another in your application. There are two ways of doing this: 

1. Add parameters to a URL

Add parameters to a URL, like this:

	<a href="greeting.asp?fname=John&lname=Smith">

Go to Welcome Page</a>


And retrieve the values like this (in the file greeting.asp):

	<%

fname=Request.querystring("fname")

lname=Request.querystring("lname")

response.write("<p>Hello " & fname & " " & lname & "!</p>")

response.write("<p>Welcome to my Web site!</p>")

%>


2. Use a form

The form passes the filled out information to the file greeting.asp when the user clicks on the Send button: 

	<form method="post" action="greeting.asp">

First Name:  <input type="text" name="fname" value="">

Last Name: <input type="text" name="lname" value="">

<input type="submit" value="Send">

</form>


Retrieve the values like this (in the file greeting.asp):

	<%

fname=Request.form("fname")

lname=Request.form("lname")

response.write("<p>Hello " & fname & " " & lname & "!</p>")

response.write("<p>Welcome to my Web site!</p>")

%>




ASP Session Object



The Session object is used to store information about, or change settings for a user session. Variables stored in the Session object hold information about one single user, and are available to all pages in one application. 



The Session object 

When you are working with an application, you open it, do some changes and then you close it. This is much like a Session. The computer knows who you are. It knows when you start the application and when you end. But on the internet there is one problem: the web server does not know who you are and what you do because the HTTP address doesn't maintain state. 

ASP solves this problem by creating a unique cookie for each user. The cookie is sent to the client and it contains information that identifies the user. This interface is called the Session object. 

The Session object is used to store information about, or change settings for a user session. Variables stored in the Session object hold information about one single user, and are available to all pages in one application. Common information stored in session variables are name, id, and preferences. The server creates a new Session object for each new user, and destroys the Session object when the session expires. 



When does a Session Start?

A session can start when: 

· A new user requests a .asp file, and the Global.asa file includes a Session_OnStart procedure 

· A value is stored in a Session variable 

· A user requests a .asp file, and the Global.asa file uses the <object> tag to instantiate an object with session scope 



When does a Session End?

A session ends if a user has not requested or refreshed a page in the ASP application for a specified period. By default, this is 20 minutes. 

If you want to set a timeout interval that is shorter or longer than the default, you can set the Timeout property. The example below sets a timeout interval of 5 minutes. 

	<%

Session.Timeout=5

%>


To end a session you can also use the Abandon method. 

	<%

Session.Abandon

%>


Note: The main problem with sessions is WHEN they should end. We do not know if the user's last request was the final one or not. So we do not know how long to keep the sessions alive. Waiting too long uses up resources on the server. But if the session is deleted too fast you risk that the user is coming back and the server has deleted all the information, so the user has to start all over again. Finding the right timeout interval can be difficult. 

Tip: If you are using session variables, store SMALL amounts of data in them. 



Store and Retrieve Variable Values 

The most important thing about the Session object is that you can store variables in it, like this: 

	<% 

Session("username")="Hege"

Session("age")=24

%>


The example above will set the Session variable username to Hege and the Session variable age to 24.

When the value is stored in a session variable it can be reached from any page in the ASP application by using: Session("username"):

	Welcome <%Response.Write(Session("username"))%>


The line above returns: "Welcome Hege". 

You can also store user preferences in the Session object, and then access that preference to choose what page to return to the user.  

The example below specifies a text-only version of the page if the user has a low screen resolution: 

	<%If Session("screenres")="low" Then%> 

  This is the text version of the page

<%Else%> 

  This is the multimedia version of the page

<%End If%>




Remove Variable Values 

The Contents collection contains all the variables that have been created and stored in a session. By using the Remove method, you can remove a variable from a session. 

The example below removes an item, in this case a session variable named "sale": 

	<%

If Session.Contents("age")<=18 then 

  Session.Contents.Remove("sale")

End If 

%>


You can also use the RemoveAll method to remove all variables in a session: 

	<%

Session.Contents.RemoveAll()

%>




Looping Through the Contents 

You can loop through the Contents collection, to see what is stored in it: 

	<%

dim i

For Each i in Session.Contents

  Response.Write(i & "<br>")

Next

%>


Result:

	username

age


If you do not know the number of items in the Contents collection, you can use the Count property:

	<%

dim i

dim j

j=Session.Contents.Count

Response.Write("Session variables:" & j)

For i=1 to j

  Response.Write(Session.Contents(i) & "<br>")

Next

%>


Result:

	Session variables: 2

Hege

24




Looping Through the Objects

You can loop through the StaticObjects collection, to see the values of all the objects stored in the Session object:

	<%

dim i

For Each i in Session.StaticObjects

  Response.Write(i & "<br>")

Next

%>




ASP Application Object



A group of ASP files that work together to perform some purpose is called an application. The Application object in ASP is used to tie these files together.



Application Object

An application on the Web is a group of ASP files. The files work together to perform some purpose. The Application object in ASP is used to tie these files together.

The Application object is used to store variables and access variables from any page, just like the Session object. The difference is that all users share ONE Application object, while with Sessions there is one Session object for each user.

The Application object should hold information that will be used by many pages in the application (like database connection information). This means that you can access the information from any page. It also means that you can change the information in one place and the new information will automatically be reflected on all pages.



Store and Retrieve Variable Values

Application variables can be accessed and changed by any page in the application.

You can create Application variables in Global.asa like this:

	<script language="vbscript" runat="server">

Sub Application_OnStart

application("vartime")=""

application("whoon")=1

End Sub

</script>


In the example above we have created two Application variables: The first is named vartime, and the second is named whoon. 

You can access the value of an Application variable like this:

	There are 

<%

Response.Write(Application("whoon"))

%> 

active connections.




Looping Through the Contents

You can loop through the "Contents" collection, to see the values of all the Application variables:

	<%

dim i

For Each i in Application.Contents

  Response.Write(i & "<br>")

Next

%>


If you don't know how many items are stored in a "Contents" collection, you can use the "Count" property:

	<%

dim i

dim j

j=Application.Contents.Count

For i=1 to j

  Response.Write(Application.Contents(i) & "<br>")

Next

%>




Looping Through the Objects

You can loop through the "StaticObjects" collection, to see the values of all the objects stored in the Application Object:

	<%

dim i

For Each i in Application.StaticObjects

  Response.Write(i & "<br>")

Next

%>




Lock and Unlock

You can lock an application with the "Lock" method. When an application is locked, the users cannot change the Application variables (other than the one currently accessing it). You can unlock an application with the "Unlock" method. This method removes the lock from the Application variable:

	<%

Application.Lock

  'do some application object operations

Application.Unlock

%>




ASP Including Files



The #include directive is used to create functions, headers, footers, or elements that will be reused on multiple pages.



The #include Directive

It is possible to insert the content of another file into an ASP file before the server executes it, with the #include directive. The #include directive is used to create functions, headers, footers, or elements that will be reused on multiple pages.



How to Use the #include Directive

Here is a file called "mypage.asp":

	<html> 

<body> 

<h3>Words of Wisdom:</h3>

<p><!--#include file="wisdom.inc"--></p> 

<h3>The time is:</h3>

<p><!--#include file="time.inc"--></p>

</body> 

</html> 


Here is the "wisdom.inc" file:

	"One should never increase, beyond what is necessary,

the number of entities required to explain anything."


Here is the "time.inc" file:

	<%

Response.Write(Time)

%>


If you look at the source code in a browser, it will look something like this: 

	<html>

<body>

<h3>Words of Wisdom:</h3>

<p>"One should never increase, beyond what is necessary,

the number of entities required to explain anything."</p>

<h3>The time is:</h3>

<p>11:33:42 AM</p>

</body>

</html>




Syntax for Including Files

To include a file into an ASP page, place the #include directive inside comment tags:

	<!--#include virtual="somefilename"-->

or

<!--#include file ="somefilename"-->


Using the Virtual Keyword

Use the virtual keyword to indicate a path beginning with a virtual directory.

For example, if a file named header.inc resides in a virtual directory named /html, the following line would insert the contents of header.inc into the file containing the line:

	<!-- #include virtual ="/html/header.inc" -->


Using the File Keyword

Use the file keyword to indicate a relative path. A relative path begins with the directory that contains the including file.

For example, if you have a file in the directory html, and the file header.inc is in html\headers, the following line would insert header.inc in your file:

	<!-- #include file ="headers\header.inc" -->


Note that the path to the included file, headers\header.inc, is relative to the including file; if the script containing this #include statement is not in the directory /html, the statement would not work.

You can also use the file keyword with the syntax (..\) to include a file from a higher-level directory.



Tips and Notes

In the sections above we have used the file extension ".inc" for the included files. Notice that if a user tries to browse a ".inc" file directly, its content will be displayed. So if your included file contains source code you do not want any users to see, it is better to use a ".asp" extension. The source code in a ".asp" file will not be visible after the interpretation.

An included file can include other files, and one ASP file can include the same file more than once.

Important: Included files are processed and inserted before the scripts are executed.

The following script will not work because ASP executes the #include directive before it assigns a value to the variable:

	<%

fname="header.inc"

%>

<!--#include file="<%=fname%>"-->


You can not open or close a script delimiter in a ".inc" file. This script will not work:

	<%

For i = 1 To n

  <!--#include file="count.inc"-->

Next

%>


But this script will work:

	<%

For i = 1 to n

%>

<!--#include file="count.inc" -->

<% Next %>




ASP The Global.asa file



In Global.asa you can specify event scripts and declare session and application objects that can be accessed by every page in an ASP application.



The Global.asa file

The Global.asa file is an optional file that can contain declarations of objects, variables, and methods that can be accessed by every page in an ASP application. All valid browser scripts (JavaScript, VBScript, JScript, PerlScript, etc.) can be used within Global.asa.

Global.asa files can contain only the following:

· Application events 

· Session events 

· <object> declarations  

· TypeLibrary declarations 

Note: The Global.asa file must be stored in the root directory of the ASP application, and each application can only have one Global.asa file. If you modify the Global.asa file at a later point, you might need to restart the server to make the changes work.



Standard Events in Global.asa

In the Global.asa file you can tell the application and session objects what to do when the application/session starts and what to do when the application/session ends. The code for this is placed into event handlers. The Global.asa file can contain four types of events:

Application_OnStart - This event occurs when the FIRST user calls the first page from an ASP application. This event occurs after the Web server is restarted or after the Global.asa file is edited. When this procedure is complete, the "Session_OnStart" procedure runs. 

Session_OnStart - This event occurs EVERY time a new user requests the first page in the ASP application.

Session_OnEnd - This event occurs EVERY time a user ends a session. A user ends a session after a page has not been requested by the user for a specified time (by default this is 20 minutes).

Application_OnEnd - This event occurs after the LAST user has ended the session. Typically, this event occurs when a Web server stops. This procedure is used to clean up settings after the Application stops, like delete records or write information to text files.

You can create a subroutine to handle each of these events in the Global.asa file:

	<script language="vbscript" runat="server">

sub Application_OnStart

......some vbscript code

end sub

sub Application_OnEnd

......some vbscript code

end sub

sub Session_OnStart

......some vbscript code

end sub

sub Session_OnEnd

......some vbscript code

end sub

</script>


Note: We do not use the ASP script delimiters, <% and %>, to insert scripts in the Global.asa file, we have to put the subroutines inside the HTML <script> element. 



<object> Declarations 

It is possible to create objects with session or application scope in Global.asa by using the <object> tag. 

This tag is outside of any <script> tags. 

Syntax 

	<object runat="server" scope=scope id=identifier

{progid="progID"|classid="classID"}>

....

</object>


	Parameter
	Description

	scope 
	Sets the scope of the object (either Session or Application) 

	id 
	Specifies a name for the object 

	ProgID 
	An id associated with a class id. The format for ProgID is [Vendor.]Component[.Version] 

Either ProgID or ClassID must be specified.

	ClassID 
	Specifies a unique id for a COM class object. 

Either ProgID or ClassID must be specified.


Examples 

	Example 1
Create an object of session scope named MyAd by using the ProgID parameter:

<object runat="server" scope="session" id="MyAd"

progid="MSWC.AdRotator">

</object>

Example 2
Creates an object of application scope named MyConnection by using the ClassID parameter:

<object runat="server" scope="application" id="MyConnection"

classid="Clsid:8AD3067A-B3FC-11CF-A560-00A0C9081C21">

</object>

Example 3
The objects declared in the Global.asa file can be used by any script in the application:

GLOBAL.ASA:

<object runat="server" scope="session" id="MyAd"

progid="MSWC.AdRotator">

</object>

You could reference the object MyAd from any page in the application:

SOME .ASP FILE:

<%=MyAd.GetAdvertisement("/banners/adrot.txt")%> 




TypeLibrary Declarations

A TypeLibrary is a container for the contents of a DLL file corresponding to a COM object. By including the call to the TypeLibrary in the GLOBAL.ASA file, the constants of that object can be accessed, and errors can be better reported by the ASP code. If your Web application relies on COM objects that have declared data types in type libraries, you can declare the type libraries in Global.asa.

Syntax 

	<!--METADATA TYPE="TypeLib"

file="filename"

uuid="typelibraryuuid"

version="versionnumber"

lcid="localeid"

-->


	Parameter
	Description

	file
	Specifies an absolute path to a type library. 

Either the file parameter or the uuid parameter is required

	uuid
	Specifies a unique identifier for the type library. 

Either the file parameter or the uuid parameter is required

	version
	Optional. Used for selecting version. If the requested version is not found, then the most recent version is used

	localeid
	Optional. The locale identifier to be used for the type library


Error Values

The server can return one of the following error messages:

	Error Code
	Description

	ASP 0222
	Invalid type library specification 

	ASP 0223
	Type library not found 

	ASP 0224
	Type library cannot be loaded

	ASP 0225
	Type library cannot be wrapped


Note: The METADATA tags can appear anywhere inside of the Global.asa file (both inside and outside the <script> tags), however it is recommended that METADATA tags appear near the top of the Global.asa file. 



Restrictions

Restrictions on what you can include in the Global.asa file: 

· You can not display text that is written in the Global.asa file. This file can't display information 

· You can not use the #include directive in Global.asa 

· You can only use Server and Application objects in the Application_OnStart and Application_OnEnd subroutines. In the Session_OnEnd subroutine, you can use Server, Application, and Session objects. In the Session_OnStart subroutine you can use any built-in object 



How to use the Subroutines

The Global.asa file is often used to initialize variables. 

The example below shows how to detect the exact time a visitor first arrives at your Web site. The time is stored in a Session variable named started, and the value of that variable can be accessed from any ASP page in the application:

	<script language="vbscript" runat="server">

sub Session_OnStart

Session("started")=now()

end sub

</script>


The Global.asa file can also be used to control page access. 

The example below shows how to redirect every new visitor to another page, in this case to a page called "newpage.asp": 

	<script language="vbscript" runat="server">

sub Session_OnStart

Response.Redirect("newpage.asp")

end sub

</script>


And you can include functions in the Global.asa file. 

In the example below the Application_OnStart subroutine occurs when the Web server starts. Then the Application_OnStart subroutine calls another subroutine named getcustomers. The getcustomers subroutine opens a database and retrieves a set of records from the customers table. The recordset is assigned to an array, where it can be accessed from any ASP page without querying the database: 

	<script language="vbscript" runat="server">

sub Application_OnStart

getcustomers

end sub

sub getcustomers 

set conn=Server.CreateObject("ADODB.Connection")

conn.Provider="Microsoft.Jet.OLEDB.4.0"

conn.Open "c:/webdata/northwind.mdb"

set rs=conn.execute("select name from customers")

Application("customers")=rs.GetRows

rs.Close

conn.Close

end sub

</script>




Global.asa Example 

In this example we will create a Global.asa file that counts the number of current visitors. 

· The Application_OnStart sets the Application variable visitors to 0 when the server starts 

· The Session_OnStart subroutine adds one to the variable visitors every time a new visitor arrives 

· The Session_OnEnd subroutine subtracts one from visitors each time this subroutine is triggered 

The Global.asa file: 

	<script language="vbscript" runat="server">

Sub Application_OnStart

Application("visitors")=0

End Sub

Sub Session_OnStart

Application.Lock

Application("visitors")=Application("visitors")+1

Application.UnLock

End Sub

Sub Session_OnEnd

Application.Lock

Application("visitors")=Application("visitors")-1

Application.UnLock

End Sub

</script>


To display the number of current visitors in an ASP file: 

	<html>

<head>

</head>

<body>

<p>

There are <%response.write(Application("visitors"))%>

online now!

</p>

</body>

</html>

